# Example 3a: Fixed Effects in General Multilevel Models for Two-Level Nested Outcomes (as estimated using restricted maximum likelihood in SAS MIXED and STATA MIXED)

The data for this example come from Snijders and Bosker (2012). In a sample of 3,305 students from 178 schools, we will be examining the extent to which student language at post-test can be predicted from school-level variables of homework amount, denomination (three types), and use of mixed-grade classes, as well as student-level predictors of verbal IQ and performance IQ. Note that this example makes use of three custom SAS macros I wrote (available in the SAS syntax file online) by which to compute total-$R^2$, pseudo-$R^2$, and regions of significance for interactions.

## SAS Syntax and Output for Data Import, Manipulation, and Description:

```
* Define global variable for file location to be replaced in code below;
* \\Client\ precedes actual path when using UIowa Virtual Desktop;
%LET filesave=C:\Dropbox\19_PSQF7375_Clustered\PSQF7375_Clustered_Example3a;

* Import example data excel data file into SAS;
PROC IMPORT DATAFILE="&filesave.\Multilevel_Analysis_Missing.xlsx"
     OUT=work.Students DBMS=XLSX REPLACE;
     SHEET="book_data";
     GETNAMES=YES; RUN;

* Data processing;
DATA work.Students; SET work.Students;
* Label variables;
  LABEL schoolID=  "School ID"
        studentID= "Student ID"
        langpost=  "Student Language"
        IQverb=    "Student Verbal IQ"
        IQperf=    "Student Performance IQ"
        denomina=  "School Denomination"
        homework=  "School Homework"
        mixedgra=  "School Mixed Grade";
* Select variables to keep;
  KEEP schoolID studentID langpost IQverb IQperf denomina homework mixedgra;
* Select complete cases;
  IF NMISS(schoolID,studentID,langpost,IQverb,IQperf,denomina,homework,mixedgra)>0 THEN
DELETE;
* Remove denomina=4,5 for small sample sizes;
  IF denomina IN(4,5) THEN DELETE; RUN;

* Z-score IQ variables for comparability as predictors;
PROC STANDARD DATA=work.Students OUT=work.StudentsZ MEAN=0 STD=1;
     VAR IQverb IQperf; RUN;
* Rename z-scored variables;
DATA work.StudentsZ; SET work.StudentsZ;
     RENAME IQverb=IQverbz IQperf=IQperfz; RUN;

* Create school-level dataset;
PROC MEANS NOPRINT DATA=work.StudentsZ; By schoolID;
     VAR langpost IQverbz IQperfz homework mixedgra;
     OUTPUT OUT=work.Schools  N(langpost)=Nperschool
     MEAN(langpost IQverbz IQperfz denomina homework mixedgra)=
          SMlangpost SMIQverbz SMIQperfz denomina homework mixedgra; RUN;
* Label new school variables;
DATA work.Schools; SET work.Schools;
  LABEL SMlangpost= "School Mean Language"
        SMIQverbz=  "School Mean Verbal IQ Z"
        SMIQperfz=  "School Mean Performance IQ Z";
  DROP _TYPE_ _FREQ_; * Remove unneeded vars; RUN;
* Merge school-level data back into student data;
DATA work.StudentsSchools; MERGE StudentsZ Schools; BY schoolID;
```

```
* Contrasts for school denomination -- initialize to missing;
  den1v2=.; den1v3=.;
  IF denomina=1 THEN DO; den1v2=0; den1v3=0; END;
  IF denomina=2 THEN DO; den1v2=1; den1v3=0; END;
  IF denomina=3 THEN DO; den1v2=0; den1v3=1; END;
  LABEL den1v2= "Denom 1 vs 2"
        den1v3= "Denom 1 vs 3";
* Center school predictors near sample mean;
  hw2=homework-2;
  SMIQverbz0=SMIQverbz-0; * Reminder to center all L2 predictors;
  SMIQperfz0=SMIQperfz-0;
  LABEL hw2= "School Homework (0=2)";
* Create group-MC student-level variables using original L2 means;
  WSlangpost=langpost-SMlangpost;
  WSIQverbz=IQverbz-SMIQverbz;
  WSIQperfz=IQperfz-SMIQperfz;
  LABEL WSlangpost= "Student Language (0=School Mean)"
        WSIQverbz=  "Student Verbal IQ Z (0=School Mean)"
        WSIQperfz=  "Student Performance IQ Z (0=School Mean)";
RUN;

TITLE "SAS School-Level Descriptives";
PROC FREQ DATA=work.Schools;
     TABLE denomina mixedgra; RUN;
PROC MEANS NDEC=2 DATA=work.Schools;
     VAR SMlangpost SMIQverbz SMIQperfz homework Nperschool; RUN; TITLE;

TITLE "SAS Student-Level Descriptives";
PROC MEANS NDEC=2 DATA=work.StudentsSchools;
     VAR langpost WSlangpost IQverbz WSIQverbz IQperfz WSIQperfz; RUN; TITLE;
```

## STATA Syntax and Output for Data Import, Manipulation, and Description:

```
// Define global variable for file location to be replaced in code below
// \\Client\ precedes actual path when using UIowa Virtual Desktop
global filesave "C:\Dropbox\19_PSQF7375_Clustered\PSQF7375_Clustered_Example3a"

// Import example data excel data file into STATA
clear  // clear memory in case a dataset is already open
import excel "$filesave\Multilevel_Analysis_Missing.xlsx", ///
       sheet("book_data") firstrow case(preserve)

// Label variables
label variable schoolID   "School ID"
label variable studentID  "Student ID"
label variable langpost   "Student Language"
label variable IQverb     "Student Verbal IQ"
label variable IQperf     "Student Performance IQ"
label variable denomina   "School Denomination"
label variable homework   "School Homework Amount"
label variable mixedgra   "School Mixed Grade"

// Select variables to keep
keep schoolID studentID langpost IQverb IQperf denomina homework mixedgra
// Select complete cases
egen nummiss = rowmiss(schoolID studentID langpost IQverb IQperf ///
               denomina homework mixedgra)
drop if nummiss>0
// Remove denomina=4,5 for small sample sizes
drop if denomina==4
drop if denomina==5

// Z-score IQ variables for comparability as predictors
egen IQverbz=std(IQverb)
egen IQperfz=std(IQperf)
```

```
// Create school-level variables
egen Nperschool=count(langpost), by(schoolID)
egen SMlangpost=mean(langpost),  by(schoolID)
egen SMIQverbz=mean(IQverbz),    by(schoolID)
egen SMIQperfz=mean(IQperfz),    by(schoolID)
// Label school-level variables
label variable SMlangpost "School Mean Language"
label variable SMIQverbz  "School Mean Verbal IQ Z"
label variable SMIQperfz  "School Mean Performance IQ Z"

// Contrasts for school denomination -- initialize to missing
gen den1v2=.
gen den1v3=.
replace den1v2=0 if denomina==1
replace den1v3=0 if denomina==1
replace den1v2=1 if denomina==2
replace den1v3=0 if denomina==2
replace den1v2=0 if denomina==3
replace den1v3=1 if denomina==3
label variable den1v2 "Denom 1 vs 2"
label variable den1v3 "Denom 1 vs 3"

// Center school predictors near sample mean
gen hw2=homework-2
gen SMIQverbz0=SMIQverbz-0 // Reminder to center all L2 predictors
gen SMIQperfz0=SMIQperfz-0
label variable hw2 "School Homework (0=2)"

// Create group-MC student-level variables using original L2 means
gen WSlangpost=langpost-SMlangpost
gen WSIQverbz=IQverbz-SMIQverbz
gen WSIQperfz=IQperfz-SMIQperfz
label variable WSlangpost "Student Language (0=School Mean)"
label variable WSIQverbz  "Student Verbal IQ Z (0=School Mean)"
label variable WSIQperfz  "Student Performance IQ Z (0=School Mean)"

display as result "STATA School-Level Descriptives"
preserve  // Save for later use, then compute school-level dataset
collapse  SMlangpost SMIQverbz SMIQperfz homework denomina mixedgra Nperschool, ///
          by(schoolID)
tabulate  denomina
tabulate  mixedgra
format    SMlangpost SMIQverbz SMIQperfz homework Nperschool %4.2f
summarize SMlangpost SMIQverbz SMIQperfz homework Nperschool, format
restore   // Go back to student-level dataset

display as result "STATA Student-Level Descriptives"
format    langpost WSlangpost IQverbz WSIQverbz IQperfz WSIQperfz %4.2f
summarize langpost WSlangpost IQverbz WSIQverbz IQperfz WSIQperfz, format
```

## STATA Old-School Listing Output (which I still use because it’s easier to paste and annotate):

```
   denomina |      Freq.     Percent        Cum.
------------+-----------------------------------
          1 |         56       31.46       31.46
          2 |         68       38.20       69.66
          3 |         54       30.34      100.00
------------+-----------------------------------
      Total |        178      100.00
```

```
   mixedgra |      Freq.     Percent        Cum.
------------+-----------------------------------
       0.00 |        116       65.17       65.17
       1.00 |         62       34.83      100.00
------------+-----------------------------------
      Total |        178      100.00

    Variable |        Obs        Mean    Std. Dev.       Min        Max
-------------+---------------------------------------------------------
  SMlangpost |        178       40.75        4.95      23.71      50.33
   SMIQverbz |        178       -0.04        0.46      -2.07       1.22
   SMIQperfz |        178       -0.03        0.35      -1.19       0.96
    homework |        178        2.41        0.53       1.33       3.67

    Variable |        Obs        Mean    Std. Dev.       Min        Max
-------------+---------------------------------------------------------
    langpost |      3,305       41.30        8.87       9.00      58.00
  WSlangpost |      3,305       -0.00        7.69     -33.55      22.70
     IQverbz |      3,305        0.00        1.00      -3.85       3.25
    WSIQverbz |      3,305       -0.00        0.92      -3.54       3.34
     IQperfz |      3,305        0.00        1.00      -2.91       2.99
   WSIQperfz |      3,305       -0.00        0.95      -2.77       2.85
  Nperschool |        178       18.57        7.37       4.00      34.00
```

---

## Model 1a for Language: Single-Level Empty Means, No Random Intercept

$$\text{Language}_{pg} = \beta_0 + e_{pg}$$

```
TITLE "SAS Model 1a for Language: Single-Level Empty Means, Random Intercept";
PROC MIXED DATA=work.StudentsSchools NOCLPRINT COVTEST NAMELEN=100 IC METHOD=REML;
     CLASS schoolID studentID;
     MODEL langpost = / SOLUTION DDFM=BW; RUN; TITLE;
```

```
            Dimensions
Covariance Parameters             1
Columns in X                      1
Columns in Z                      0
Subjects                          1
Max Obs per Subject            3305
```

This model is treating these cases as independent—one group of 3,305 students.

```
          Number of Observations
Number of Observations Read             3305
Number of Observations Used             3305
Number of Observations Not Used            0
```

This table tells you how many cases were removed due to incomplete data—make sure you pay attention to this if you are doing any model comparisons (which will need to be based on the exact same cases to be valid).

```
            Covariance Parameter Estimates
                         Standard         Z
Cov Parm      Estimate      Error     Value        Pr > Z
Residual       78.7355     1.9372     40.64        <.0001
```

This is the variance of the $e_{pg}$ residuals assuming independent observations.

```
                        Information Criteria
Neg2LogLike    Parms       AIC       AICC       HQIC        BIC       CAIC
    23810.0        1   23812.0    23812.0    23814.2    23818.1    23819.1

                  Solution for Fixed Effects
                         Standard
Effect        Estimate      Error      DF    t Value    Pr > |t|
Intercept      41.3047     0.1543    3304     267.61      <.0001
```

This is the grand mean for language, pooling over all of the observations.

```
display as result "STATA Model 1a for Language: Empty Means, Random Intercept"
mixed langpost , ///
      variance reml dfmethod(residual)
```

```
Mixed-effects REML regression                   Number of obs     =      3,305
DF method: Residual                             DF:           min =   3,304.00
                                                              avg =   3,304.00
                                                              max =   3,304.00
                                                F(0,  3304.00)    =          .
Log restricted-likelihood = -11905.013          Prob > F          =          .

------------------------------------------------------------------------------
    langpost |      Coef.   Std. Err.      t    P>|t|     [95% Conf. Interval]
-------------+----------------------------------------------------------------
       _cons |   41.30469   .1543475   267.61   0.000     41.00206    41.60732
------------------------------------------------------------------------------

------------------------------------------------------------------------------
  Random-effects Parameters  |   Estimate   Std. Err.     [95% Conf. Interval]
-----------------------------+------------------------------------------------
               var(Residual) |   78.73553   1.936868       75.0294    82.62472
------------------------------------------------------------------------------
```

This is the grand mean for language, pooling over all of the observations.

This is the variance of the $e_{pg}$ residuals assuming independent observations.

---

## Model 1b for Language: Single-Level Empty Means, Random Intercept

$$\text{Level-1: Language}_{pg} = \beta_{0g} + e_{pg}$$

$$\text{Level-2:} \quad \beta_{0g} = \gamma_{00} + U_{0g}$$

```
TITLE "SAS Model 1b for Language: Two-Level Empty Means, Random Intercept";
PROC MIXED DATA=work.StudentsSchools NOCLPRINT COVTEST NAMELEN=100 IC METHOD=REML;
     CLASS schoolID studentID; * OUTPM=Save predicted outcomes;
     MODEL langpost = / SOLUTION DDFM=Satterthwaite OUTPM=PredEmpty;
     RANDOM INTERCEPT / VCORR TYPE=UN SUBJECT=schoolID; * VCORR gives ICC;
     ODS OUTPUT CovParms=CovEmpty; * Save variances; RUN; TITLE;
```

Changes to the syntax: **RANDOM** lists all effects to have random variances across the unit of analysis provided by SUBJECT. TYPE=UN means unstructured—this should be left alone. On MODEL, denominator degrees of freedom (DDFM) now uses **Satterthwaite**, a good general MLM option (Kenward-Roger is better for small samples).

OUTPM requested a dataset called "PredEmpty" that will contain a column "pred" as the model-predicted outcome. Right now, this will be a constant (the mean), but we will eventually use this to compute total-$R^2$. ODS OUTPUT saves the variance parameters to a dataset called "CovEmpty" that will be used to compute pseudo-$R^2$ later.

```
Covariance Parameters            2
Columns in X                     1
Columns in Z per Subject         1
Subjects                       178
Max Obs per Subject             34
```

Covariance parameters refers to the total number of parameters in the model for the variance. Columns in X refers to the number of fixed effects in the design matrix (i.e., including redundant columns for categorical predictors. Columns in Z refers to the number of random effects.

This model is now treating these cases as nested within schools: 178 schools with up 34 students each (mean ~19, range = 4 to 34).

```
                     Iteration History
Iteration    Evaluations    -2 Res Log Like      Criterion
        0              1     23810.02512060
        1              2     23373.80532705     0.00005932
        2              1     23373.24829101     0.00000122
        3              1     23373.23762514     0.00000000
```

For your homework, report the last −2LL value from this table to obtain two digits past the decimal for enough precision.

```
Estimated V Correlation Matrix for schoolID 1 (truncated to save space)
Row      Col1      Col2
  1     1.0000    0.2282
  2     0.2282    1.0000
```

Intraclass Correlation as provided by VCORR:

$$ICC = \frac{\tau^2_{U_0}}{\tau^2_{U_0} + \sigma^2_e} = \frac{18.5134}{18.5134 + 92.6200} = .2282$$

```
                  Covariance Parameter Estimates
                                   Standard        Z
Cov Parm      Subject     Estimate    Error     Value      Pr > Z
UN(1,1)       schoolID     18.5134    2.4694     7.50      <.0001 → L2 random intercept variance
Residual                   62.6200    1.5866    39.47      <.0001 → L1 residual variance
```

```
  Null Model Likelihood Ratio Test
    DF     Chi-Square      Pr > ChiSq
     1         436.79          <.0001
```

This Null Model LRT is the difference in −2LL (−2ΔLL) between this model and the same model without any additional parameters in the model for the variance besides residual variance. Here, this is the significance test of the ICC, which in turn provides an effect size for the amount of dependency attributed to school mean differences in language.

```
                            Information Criteria
Neg2LogLike     Parms        AIC        AICC        HQIC         BIC        CAIC
    23373.2         2    23377.2     23377.2     23379.8     23383.6     23385.6
```

REML's parm count (2) excludes the fixed effects.

```
                      Solution for Fixed Effects
                            Standard
Effect         Estimate        Error       DF     t Value     Pr > |t|
Intercept       40.8960       0.3553      164      115.11       <.0001
```

The intercept is now the grand mean of the school means, which will differ from the overall grand mean whenever level-2 units have different level-1 sizes (unbalanced).

---

```
display as result "STATA Model 1b for Language: Empty Means, Random Intercept"
mixed langpost , ///
          || schoolID: , variance reml covariance(unstructured) dfmethod(satterthwaite),
        estat icc  // Compute Intraclass Correlation
```

Changes to the syntax: || denotes the random part of the model, which currently includes just an intercept variance across schools (included by default). Covariance(unstructured) is the analog to TYPE=UN means unstructured—this should be left alone (although it doesn't matter here). On MODEL, denominator degrees of freedom (DDFM) now uses **Satterthwaite**, a good general MLM option (Kenward-Roger is better for small samples).

```
Mixed-effects REML regression                   Number of obs     =      3,305
Group variable: schoolID                        Number of groups  =        178

                                                Obs per group:
                                                              min =          4
                                                              avg =       18.6
                                                              max =         34
DF method: Satterthwaite                        DF:           min =     175.69
                                                              avg =     175.69
                                                              max =     175.69

                                                F(0,       0.00)  =          .
Log restricted-likelihood = -11686.619          Prob > F          =          .

------------------------------------------------------------------------------
    langpost |      Coef.   Std. Err.      t    P>|t|     [95% Conf. Interval]
-------------+----------------------------------------------------------------
       _cons |   40.89593   .3553269   115.09   0.000     40.19467    41.59718
------------------------------------------------------------------------------
```

The intercept is now the grand mean of the school means, which will differ from the overall grand mean whenever level-2 units have different level-1 sizes (are unbalanced).

```
---------------------------------------------------------------------------
  Random-effects Parameters  |   Estimate   Std. Err.     [95% Conf. Interval]
-----------------------------+---------------------------------------------
schoolID: Identity           |
                  var(_cons) |    18.5189   2.470581      14.25797    24.0532 → L2 random int var
-----------------------------+---------------------------------------------
               var(Residual) |   62.61918   1.586618      59.58542   65.80741 → L1 residual var
---------------------------------------------------------------------------
LR test vs. linear model: chibar2(01) = 436.79        Prob >= chibar2 = 0.0000
```

This Null Model LRT is the difference in −2LL between this model and the same model without any additional parameters in the model for the variance besides residual variance. Here, this is the significance test of the ICC, which in turn provides an effect size for the amount of dependency attributed to school mean differences in language.

```
Intraclass correlation
------------------------------------------------------------------------------
                       Level |        ICC   Std. Err.     [95% Conf. Interval]
-----------------------------+------------------------------------------------
                    schoolID |   .2282393   .0241629      .1843516    .2790014
------------------------------------------------------------------------------
```

**Design effect** using mean #students per school: = 1 + ((*n* – 1) * ICC) → 1 + [(18−1)*.228] = **4.876**

**Effective sample size:** $N_{effective}$ = (#Total Obs) / Design Effect → 3,305 / 4.876 = **677!!!**

**This means that our power to detect level-1 effects will be approximately that of an independent sample of 677 students—only if the ICC were 0 would we have power for level-1 effects based on the actual number of students. Power for level-2 effects is based on the number of schools (178).**

---

## Model 2a: Add 3 School Predictors

$$\text{Level-1: Language}_{pg} = \beta_{0g} + e_{pg}$$

$$\text{Level-2: } \beta_{0g} = \gamma_{00} + \gamma_{01}(\text{Homework}_g - 2) + \gamma_{02}(\text{MixedGrade}_g) + \gamma_{03}(\text{Denom1v2}_g) + \gamma_{04}(\text{Denom1v3}_g) + U_{0g}$$

```
TITLE "SAS Model 2a: Add 3 School Predictors";
PROC MIXED DATA=work.StudentsSchools NOCLPRINT COVTEST NAMELEN=100 IC METHOD=REML;
     CLASS schoolID studentID;
     MODEL langpost = hw2 mixedgra den1v2 den1v3
                      / SOLUTION DDFM=Satterthwaite OUTPM=PredHMD;
     RANDOM INTERCEPT / TYPE=UN SUBJECT=schoolID;
     ESTIMATE "Denomination 2 vs 3"                        den1v2 -1  den1v3 1;
     CONTRAST "Test of Model Total-R2" hw2 1, mixedgra 1, den1v2  1, den1v3 1;
     CONTRAST "Test of Change in R2 after HW" mixedgra 1, den1v2  1, den1v3 1;
     CONTRAST "Test of Omnibus Denomination"               den1v2  1, den1v3 1;
     ODS OUTPUT CovParms=CovHMD; RUN; TITLE;

display as result "STATA Model 2a: Add 3 School Predictors"
mixed langpost c.hw2 c.mixedgra c.den1v2 c.den1v3, ///
        || schoolID: , variance reml covariance(unstructured) dfmethod(satterthwaite),
       test (c.mixedgra=0) (c.den1v2=0) (c.den1v3=0), small // Test Change in R2 after HW
       test                (c.den1v2=0) (c.den1v3=0), small // Test Omnibus Denomination
       lincom c.den1v2*-1 + c.den1v3*1, small // Denomination 2 vs 3
       predict PredHMD, xb
       corr langpost PredHMD
       display as result r(rho)^2 // total R2
```

**SAS Results:**

Covariance Parameter Estimates

| Cov Parm | Subject | Estimate | Standard Error | Z Value | Pr > Z |
|---|---|---|---|---|---|
| UN(1,1) | schoolID | 15.9505 | 2.2126 | 7.21 | <.0001 |
| Residual | | 62.6202 | 1.5866 | 39.47 | <.0001 |

Information Criteria

| Neg2LogLike | Parms | AIC | AICC | HQIC | BIC | CAIC |
|---|---|---|---|---|---|---|
| **23343.0** | 2 | 23347.0 | 23347.0 | 23349.6 | 23353.4 | 23355.4 |

Solution for Fixed Effects

| Effect | Estimate | Standard Error | DF | t Value | Pr > \|t\| |
|---|---|---|---|---|---|
| Intercept | 40.0024 | 0.7248 | 157 | 55.19 | <.0001 |
| hw2 | 0.8307 | 0.6403 | 156 | 1.30 | 0.1964 |
| mixedgra | -2.0141 | 0.7336 | 174 | -2.75 | 0.0067 |
| den1v2 | 2.7418 | 0.8142 | 158 | 3.37 | 0.0010 |
| den1v3 | 0.4717 | 0.8584 | 162 | 0.55 | 0.5834 |

## Interpret each fixed effect:

Intercept $\gamma_{00}$ =

HW2 $\gamma_{01}$ =

MixedGra $\gamma_{02}$ =

Den1v2 $\gamma_{03}$ =

Den1v3 $\gamma_{04}$ =

Estimates

| Label | Estimate | Standard Error | DF | t Value | Pr > \|t\| |
|---|---|---|---|---|---|
| Denomination 2 vs 3 | -2.2701 | 0.8265 | 159 | -2.75 | 0.0067 |

Contrasts

| Label | Num DF | Den DF | F Value | Pr > F |
|---|---|---|---|---|
| Test of Model Total-R2 | 4 | 162 | 6.68 | <.0001 |
| Test of Change in R2 after HW | 3 | 164 | 8.28 | <.0001 |
| Test of Omnibus Denomination | 2 | 159 | 6.61 | 0.0017 |

```
* Calculate PseudoR2 relative to two-level empty means model 1b;
  %PseudoR2(NCov=2, CovFewer=CovEmpty, CovMore=CovHMD);
```

PseudoR2 (% Reduction) for CovEmpty vs. CovHMD

| Name | CovParm | Subject | Estimate | StdErr | ZValue | ProbZ | PseudoR2 |
|---|---|---|---|---|---|---|---|
| CovEmpty | UN(1,1) | schoolID | 18.5134 | 2.4694 | 7.50 | <.0001 | . |
| CovEmpty | Residual | | 62.6200 | 1.5866 | 39.47 | <.0001 | . |
| CovHMD | UN(1,1) | schoolID | 15.9505 | 2.2126 | 7.21 | <.0001 | 0.13843 |
| CovHMD | Residual | | 62.6202 | 1.5866 | 39.47 | <.0001 | -0.00000 |

```
* Calculate TotalR2 relative to two-level empty means model 1b;
  %TotalR2(DV=langpost, PredFewer=PredEmpty, PredMore=PredHMD);
```

Total R2 (% Reduction) for PredEmpty vs. PredHMD

| Name | Pred Corr | TotalR2 | Total R2Diff |
|---|---|---|---|
| PredEmpty | 0.00000 | 0.000000 | . |
| PredHMD | 0.17925 | 0.032132 | 0.032132 |

**$R^2$ Results:** The four fixed effects of our three level-2 predictors accounted for 13.84% of the level-2 random intercept variance, which was 3.231% of the total language variance (which can be approximated by .1384*.2283 = 3.159% using the ICC from the empty model).

## Model 2a: Add Interaction of Mixed Grade by Denomination

$$\text{Level-1: Language}_{pg} = \beta_{0g} + e_{pg}$$

$$\text{Level-2: } \beta_{0g} = \gamma_{00} + \gamma_{01}(\text{Homework}_g - 2) + \gamma_{02}(\text{MixedGrade}_g) + \gamma_{03}(\text{Denom1v2}_g) + \gamma_{04}(\text{Denom1v3}_g)$$
$$+ \gamma_{04}(\text{MixedGrade}_g)(\text{Denom1v2}_g) + \gamma_{05}(\text{MixedGrade}_g)(\text{Denom1v3}_g) + U_{0g}$$

```
TITLE "SAS Model 2b: Add Interaction of Mixed Grade by Denomination";
PROC MIXED DATA=work.StudentsSchools NOCLPRINT COVTEST NAMELEN=100 IC METHOD=REML;
     CLASS schoolID studentID;
     MODEL langpost = hw2 mixedgra den1v2 den1v3 mixedgra*den1v2 mixedgra*den1v3
                      / SOLUTION DDFM=Satterthwaite OUTPM=PredMGxD;
     RANDOM INTERCEPT / TYPE=UN SUBJECT=schoolID;
     CONTRAST "Omnibus Test of MixedGra by Den Interaction (Change in R2)"
              mixedgra*den1v2 1, mixedgra*den1v3 1;
     ESTIMATE "Denomination 2 vs 3" den1v2 -1 den1v3 1;
 * Simple effects of mixed grade per denomination;
 ESTIMATE "MixedGra Effect for Den=1" mixedgra 1 mixedgra*den1v2 0 mixedgra*den1v3 0;
 ESTIMATE "MixedGra Effect for Den=2" mixedgra 1 mixedgra*den1v2 1 mixedgra*den1v3 0;
 ESTIMATE "MixedGra Effect for Den=3" mixedgra 1 mixedgra*den1v2 0 mixedgra*den1v3 1;
 * Simple effects of denomination per MixedGra;
 ESTIMATE "Den 1v2 for MixedGra=0" den1v2  1 den1v3 0 mixedgra*den1v2  0 mixedgra*den1v3 0;
 ESTIMATE "Den 1v3 for MixedGra=0" den1v2  0 den1v3 1 mixedgra*den1v2  0 mixedgra*den1v3 0;
 ESTIMATE "Den 2v3 for MixedGra=0" den1v2 -1 den1v3 1 mixedgra*den1v2  0 mixedgra*den1v3 0;
 ESTIMATE "Den 1v2 for MixedGra=1" den1v2  1 den1v3 0 mixedgra*den1v2  0 mixedgra*den1v3 0;
 ESTIMATE "Den 1v3 for MixedGra=1" den1v2  0 den1v3 1 mixedgra*den1v2  0 mixedgra*den1v3 1;
 ESTIMATE "Den 2v3 for MixedGra=1" den1v2 -1 den1v3 1 mixedgra*den1v2 -1 mixedgra*den1v3 1;
 * Differences in simple effects -- interaction contrasts;
 ESTIMATE "MixedGra Effect for Den=1 vs 2" mixedgra*den1v2  1 mixedgra*den1v3 0;
 ESTIMATE "MixedGra Effect for Den=1 vs 3" mixedgra*den1v2  0 mixedgra*den1v3 1;
 ESTIMATE "MixedGra Effect for Den=2 vs 3" mixedgra*den1v2 -1 mixedgra*den1v3 1;
 ODS OUTPUT CovParms=CovMGbxD; RUN; TITLE;
```

```
display as result "STATA Model 2b: Add Interaction of Mixed Grade by Denomination"
mixed langpost c.hw2 c.mixedgra c.den1v2 c.den1v3 c.mixedgra#c.den1v2 c.mixedgra#c.den1v3, ///
         || schoolID: , variance reml covariance(unstructured) dfmethod(satterthwaite),
test (c.mixedgra#c.den1v2=0) (c.mixedgra#c.den1v3=0), small // Test Omnibus Interaction
// Simple effects of mixed grade per denomination
lincom c.mixedgra*1 + c.mixedgra#c.den1v2*0  + c.mixedgra#c.den1v3*0, small // MG:Den=1
lincom c.mixedgra*1 + c.mixedgra#c.den1v2*1  + c.mixedgra#c.den1v3*0, small // MG:Den=2
lincom c.mixedgra*1 + c.mixedgra#c.den1v2*0  + c.mixedgra#c.den1v3*1, small // MG:Den=3
// Simple effects of denomination per mixed grade
lincom c.den1v2*1  + c.den1v3*0 + c.mixedgra#c.den1v2*0  + c.mixedgra#c.den1v3*0, small // D1v2: MG=0
lincom c.den1v2*0  + c.den1v3*1 + c.mixedgra#c.den1v2*0  + c.mixedgra#c.den1v3*0, small // D1v3: MG=0
lincom c.den1v2*-1 + c.den1v3*1 + c.mixedgra#c.den1v2*0  + c.mixedgra#c.den1v3*0, small // D2v3: MG=0
lincom c.den1v2*1  + c.den1v3*0 + c.mixedgra#c.den1v2*1  + c.mixedgra#c.den1v3*0, small // D1v2: MG=1
lincom c.den1v2*0  + c.den1v3*1 + c.mixedgra#c.den1v2*0  + c.mixedgra#c.den1v3*1, small // D1v3: MG=1
lincom c.den1v2*-1 + c.den1v3*1 + c.mixedgra#c.den1v2*-1 + c.mixedgra#c.den1v3*1, small // D2v3: MG=1
// Differences in simple effects -- interaction contrasts
lincom c.mixedgra#c.den1v2*1  + c.mixedgra#c.den1v3*0, small // MG:Den=1v2
lincom c.mixedgra#c.den1v2*0  + c.mixedgra#c.den1v3*1, small // MG:Den=1v3
lincom c.mixedgra#c.den1v2*-1 + c.mixedgra#c.den1v3*1, small // MG:Den=2v3
predict PredMGxD, xb
corr langpost PredMGxD
display as result r(rho)^2 // total R2
```

### SAS Results:

Covariance Parameter Estimates

| Cov Parm | Subject | Estimate | Standard Error | Z Value | Pr > Z |
|---|---|---|---|---|---|
| UN(1,1) | schoolID | 15.8630 | 2.2178 | 7.15 | <.0001 |
| Residual | | 62.6251 | 1.5869 | 39.46 | <.0001 |

Information Criteria

| Neg2LogLike | Parms | AIC | AICC | HQIC | BIC | CAIC |
|---|---|---|---|---|---|---|
| **23334.8** | 2 | 23338.8 | 23338.8 | 23341.4 | 23345.2 | 23347.2 |

Solution for Fixed Effects

| Effect | Estimate | Standard Error | DF | t Value | Pr > \|t\| |
|---|---|---|---|---|---|
| Intercept | 39.9590 | 0.8373 | 146 | 47.73 | <.0001 |
| hw2 | 0.7927 | 0.6433 | 154 | 1.23 | 0.2197 |
| mixedgra | -1.8665 | 1.2383 | 168 | -1.51 | 0.1336 |
| den1v2 | 2.4937 | 0.9861 | 145 | 2.53 | 0.0125 |
| den1v3 | 1.0693 | 1.0788 | 145 | 0.99 | 0.3233 |
| mixedgra*den1v2 | 1.1955 | 1.7797 | 171 | 0.67 | 0.5027 |
| mixedgra*den1v3 | -1.6545 | 1.7676 | 171 | -0.94 | 0.3506 |

**Interpret each fixed effect:**

MixedGra $\gamma_{02}$ =

Den1v2 $\gamma_{03}$ =

Den1v3 $\gamma_{04}$ =

MixedGra*Den1v2 $\gamma_{05}$ =

MixedGra*Den1v2 $\gamma_{06}$ =

Estimates

| Label | Estimate | Standard Error | DF | t Value | Pr > \|t\| |
|---|---|---|---|---|---|
| Denomination 2 vs 3 | -1.4244 | 0.9818 | 145 | -1.45 | 0.1490 |
| MixedGra Effect for Den=1 | -1.8665 | 1.2383 | 168 | -1.51 | 0.1336 |
| MixedGra Effect for Den=2 | -0.6710 | 1.2779 | 173 | -0.53 | 0.6002 |
| MixedGra Effect for Den=3 | -3.5209 | 1.2751 | 174 | -2.76 | 0.0064 |
| Den 1v2 for MixedGra=0 | 2.4937 | 0.9861 | 145 | 2.53 | 0.0125 |
| Den 1v3 for MixedGra=0 | 1.0693 | 1.0788 | 145 | 0.99 | 0.3233 |
| Den 2v3 for MixedGra=0 | -1.4244 | 0.9818 | 145 | -1.45 | 0.1490 |
| Den 1v2 for MixedGra=1 | 2.4937 | 0.9861 | 145 | 2.53 | 0.0125 |
| Den 1v3 for MixedGra=1 | -0.5852 | 1.4036 | 191 | -0.42 | 0.6772 |
| Den 2v3 for MixedGra=1 | -4.2744 | 1.5198 | 188 | -2.81 | 0.0054 |
| MixedGra Effect for Den=1 vs 2 | 1.1955 | 1.7797 | 171 | 0.67 | 0.5027 |
| MixedGra Effect for Den=1 vs 3 | -1.6545 | 1.7676 | 171 | -0.94 | 0.3506 |
| MixedGra Effect for Den=2 vs 3 | -2.8500 | 1.8054 | 174 | -1.58 | 0.1162 |

Contrasts

| Label | Num DF | Den DF | F Value | Pr > F |
|---|---|---|---|---|
| Omnibus Test of MixedGra by Den Interaction (Change in R2) | 2 | 172 | 1.26 | 0.2865 |

```
* Calculate PseudoR2 relative to two-level empty means model 1b;
  %PseudoR2(NCov=2, CovFewer=CovEmpty, CovMore=CovMGbxD);
```

PseudoR2 (% Reduction) for CovEmpty vs. CovMGbxD

| Name | CovParm | Subject | Estimate | StdErr | ZValue | ProbZ | PseudoR2 |
|---|---|---|---|---|---|---|---|
| CovEmpty | UN(1,1) | schoolID | 18.5134 | 2.4694 | 7.50 | <.0001 | . |
| CovEmpty | Residual | | 62.6200 | 1.5866 | 39.47 | <.0001 | . |
| CovMGbxD | UN(1,1) | schoolID | 15.8630 | 2.2178 | 7.15 | <.0001 | 0.14316 |
| CovMGbxD | Residual | | 62.6251 | 1.5869 | 39.46 | <.0001 | -0.00008 |

```
* Calculate PseudoR2 relative to L2 main effects model 2a;
  %PseudoR2(NCov=2, CovFewer=CovHMD, CovMore=CovMGbxD);
```

PseudoR2 (% Reduction) for CovHMD vs. CovMGbxD

| Name | CovParm | Subject | Estimate | StdErr | ZValue | ProbZ | PseudoR2 |
|---|---|---|---|---|---|---|---|
| CovHMD | UN(1,1) | schoolID | 15.9505 | 2.2126 | 7.21 | <.0001 | . |
| CovHMD | Residual | | 62.6202 | 1.5866 | 39.47 | <.0001 | . |
| CovMGbxD | UN(1,1) | schoolID | 15.8630 | 2.2178 | 7.15 | <.0001 | 0.005487836 |
| CovMGbxD | Residual | | 62.6251 | 1.5869 | 39.46 | <.0001 | -.000077981 |

```
* Calculate TotalR2 relative to L2 main effects model 2a;
  %TotalR2(DV=langpost, PredFewer=PredHMD, PredMore=PredMGxD);
```

Total R2 (% Reduction) for PredHMD vs. PredMGxD

| Name | Pred Corr | TotalR2 | Total R2Diff |
|---|---|---|---|
| PredHMD | 0.17925 | 0.032132 | . |
| PredMGxD | 0.18830 | 0.035458 | .003326123 |

**Interpret the results for pseudo-$R^2$ and total-$R^2$:**

---

## Model 1b (Empty Means, Random Intercept) for Student-Level IQverb and IQperf to Get ICCs

```
TITLE "SAS Model 1b for IQverb: Two-Level Empty Means, Random Intercept";
PROC MIXED DATA=work.StudentsSchools NOCLPRINT COVTEST NAMELEN=100 IC METHOD=REML;
     CLASS schoolID studentID;
     MODEL IQverbz = / SOLUTION DDFM=Satterthwaite;
     RANDOM INTERCEPT / VCORR TYPE=UN SUBJECT=schoolID; RUN; TITLE;

display as result "STATA Model 1b for IQverb: Empty Means, Random Intercept"
mixed IQverbz , ///
          || schoolID: , variance reml covariance(unstructured) dfmethod(satterthwaite),
       estat icc  // Compute Intraclass Correlation
```

### SAS Results:

Estimated V Correlation Matrix for schoolID 1 (truncated to save space)

| Row | Col1 | Col2 |
|---|---|---|
| 1 | 1.0000 | 0.2282 |
| 2 | **0.1201** | 1.0000 |

Intraclass Correlation as provided by VCORR:

$$\text{ICC} = \frac{\tau_{U_0}^2}{\tau_{U_0}^2 + \sigma_e^2} = \frac{0.1221}{0.1221 + 0.8941} = .1201$$

Covariance Parameter Estimates

| Cov Parm | Subject | Estimate | Standard Error | Z Value | Pr > Z |
|---|---|---|---|---|---|
| UN(1,1) | schoolID | 0.1221 | 0.02002 | 6.10 | <.0001 |
| Residual | | 0.8941 | 0.02269 | 39.40 | <.0001 |

The LRT indicated the ICC was significant, $-2\Delta LL(1) = 154.28$, $p < .0001$.

```
TITLE "SAS Model 1b for IQperf: Two-Level Empty Means, Random Intercept";
PROC MIXED DATA=work.StudentsSchools NOCLPRINT COVTEST NAMELEN=100 IC METHOD=REML;
     CLASS schoolID studentID;
     MODEL IQperfz = / SOLUTION DDFM=Satterthwaite;
     RANDOM INTERCEPT / VCORR TYPE=UN SUBJECT=schoolID; RUN; TITLE;
display as result "STATA Model 1b for IQperf: Empty Means, Random Intercept"
mixed IQperfz , ///
         || schoolID: , variance reml covariance(unstructured) dfmethod(satterthwaite),
       estat icc  // Compute Intraclass Correlation
```

**SAS Results:**

```
Estimated V Correlation Matrix for schoolID 1 (truncated to save space)
Row       Col1        Col2
   1     1.0000      0.2282
   2     0.04847     1.0000
```

Intraclass Correlation as provided by VCORR:
$$ICC = \frac{\tau^2_{U_0}}{\tau^2_{U_0} + \sigma^2_e} = \frac{0.04854}{0.04854 + 0.9528} = .04847$$

```
                  Covariance Parameter Estimates
                                     Standard         Z
Cov Parm     Subject      Estimate      Error     Value      Pr > Z
UN(1,1)      schoolID      0.04854    0.01124      4.32      <.0001
Residual                    0.9528    0.02411     39.53      <.0001
```

The LRT indicated the ICC was significant, $-2\Delta LL(1) = 45.20$, $p < .0001$.

---

## Model 3a: Remove Mixed Grade by Denomination, Add L2 Group Mean and Group-MC L1 Student Verbal IQ

$$\text{Level-1: } \text{Language}_{pg} = \beta_{0g} + \beta_{1g}(\text{IQverb}_{pg} - \overline{\text{IQverb}}_g) + e_{pg}$$

$$\text{Level-2: } \beta_{0g} = \gamma_{00} + \gamma_{01}(\text{Homework}_g - 2) + \gamma_{02}(\text{MixedGrade}_g) + \gamma_{03}(\text{Denom1v2}_g) + \gamma_{04}(\text{Denom1v3}_g) + \gamma_{05}(\overline{\text{IQverb}}_g) + U_{0g}$$

$$\beta_{1g} = \gamma_{10}$$

```
TITLE "SAS Model 3a: Remove Mixed Grade by Denom, Add L2 and L1 Student Verbal IQ";
PROC MIXED DATA=work.StudentsSchools NOCLPRINT COVTEST NAMELEN=100 IC METHOD=REML;
     CLASS schoolID studentID;
     MODEL langpost = hw2 mixedgra den1v2 den1v3 SMIQverbz0 WSIQverbz
                  / SOLUTION DDFM=Satterthwaite OUTPM=PredIQverb;
     RANDOM INTERCEPT / TYPE=UN SUBJECT=schoolID;
     CONTRAST "Omnibus Test of Verbal IQ (Change in R2)" SMIQverbz0 1, WSIQverbz 1;
     ESTIMATE "Denomination 2 vs 3" den1v2 -1 den1v3 1;
     ESTIMATE "L2 Contextual Effect of Verbal IQ" SMIQverbz0 1 WSIQverbz -1;
     ODS OUTPUT CovParms=CovIQverb; RUN; TITLE;


display as result "STATA Model 3a: Remove MixGra by Denom, Add L2 and L1 Student Verbal IQ"
mixed langpost c.hw2 c.mixedgra c.den1v2 c.den1v3 c.SMIQverbz0 c.WSIQverbz, ///
         || schoolID: , variance reml covariance(unstructured) dfmethod(satterthwaite),
       test (c.SMIQverbz0=0) (c.WSIQverbz=0), small  // Test Omnibus Verbal IQ
       lincom c.den1v2*-1 + c.den1v3*1, small        // Denomination 2 vs 3
       lincom c.SMIQverbz0*1 + c.WSIQverbz*-1, small // L1 Contextual Effect of Verbal IQ
       predict PredIQverb, xb
       corr langpost PredIQverb
       display as result r(rho)^2 // total R2
```

## SAS Results:

Covariance Parameter Estimates

| Cov Parm | Subject | Estimate | Standard Error | Z Value | Pr > Z |
|---|---|---|---|---|---|
| UN(1,1) | schoolID | 7.9235 | 1.1341 | 6.99 | <.0001 |
| Residual | | 40.5184 | 1.0252 | 39.52 | <.0001 |

Information Criteria

| Neg2LogLike | Parms | AIC | AICC | HQIC | BIC | CAIC |
|---|---|---|---|---|---|---|
| 21870.4 | 2 | 21874.4 | 21874.4 | 21877.0 | 21880.8 | 21882.8 |

Solution for Fixed Effects

| Effect | Estimate | Standard Error | DF | t Value | Pr > \|t\| |
|---|---|---|---|---|---|
| Intercept | 40.8784 | 0.5294 | 164 | 77.22 | <.0001 |
| hw2 | 0.1425 | 0.4666 | 163 | 0.31 | 0.7605 |
| mixedgra | -1.0608 | 0.5390 | 183 | -1.97 | 0.0506 |
| den1v2 | 1.6246 | 0.5965 | 166 | 2.72 | 0.0072 |
| den1v3 | -0.3620 | 0.6263 | 170 | -0.58 | 0.5640 |
| SMIQverbz0 | 7.2988 | 0.5689 | 192 | 12.83 | <.0001 |
| WSIQverbz | 4.9720 | 0.1206 | 3124 | 41.21 | <.0001 |

## Interpret each fixed effect:

SMIQverbz0 $\gamma_{05}$ =

WSIQverbz $\gamma_{10}$ =

Estimates

| Label | Estimate | Standard Error | DF | t Value | Pr > \|t\| |
|---|---|---|---|---|---|
| Denomination 2 vs 3 | -1.9866 | 0.5992 | 165 | -3.32 | 0.0011 |
| L2 Contextual Effect of Verbal IQ | 2.3268 | 0.5816 | 210 | 4.00 | <.0001 |

Contrasts

| Label | Num DF | Den DF | F Value | Pr > F |
|---|---|---|---|---|
| Omnibus Test of Verbal IQ (Change in R2) | 2 | 361 | 931.48 | <.0001 |

```
* Calculate PseudoR2 relative to L2 main effects model 2a;
  %PseudoR2(NCov=2, CovFewer=CovHMD, CovMore=CovIQverb);
```

PseudoR2 (% Reduction) for CovHMD vs. CovIQverb

| Name | CovParm | Subject | Estimate | StdErr | ZValue | ProbZ | PseudoR2 |
|---|---|---|---|---|---|---|---|
| CovHMD | UN(1,1) | schoolID | 15.9505 | 2.2126 | 7.21 | <.0001 | . |
| CovHMD | Residual | | 62.6202 | 1.5866 | 39.47 | <.0001 | . |
| CovIQverb | UN(1,1) | schoolID | 7.9235 | 1.1341 | 6.99 | <.0001 | 0.50324 |
| CovIQverb | Residual | | 40.5184 | 1.0252 | 39.52 | <.0001 | 0.35295 |

```
* Calculate TotalR2 relative to L2 main effects model 2a;
  %TotalR2(DV=langpost, PredFewer=PredHMD, PredMore=PredIQverb);
```

Total R2 (% Reduction) for PredHMD vs. PredIQverb

| Name | Pred Corr | TotalR2 | Total R2Diff |
|---|---|---|---|
| PredHMD | 0.17925 | 0.03213 | . |
| PredIQverb | 0.62673 | 0.39279 | 0.36065 |

## Interpret the results for pseudo-$R^2$ and total-$R^2$:

## Model 3b: Add L2 Group Mean and L1 Group-MC Student Performance IQ

$$\text{Level-1: Language}_{pg} = \beta_{0g} + \beta_{1g}(\text{IQverb}_{pg} - \overline{\text{IQverb}}_{g}) + \beta_{2g}(\text{IQperf}_{pg} - \overline{\text{IQperf}}_{g}) + e_{pg}$$

$$\text{Level-2: } \beta_{0g} = \gamma_{00} + \gamma_{01}(\text{Homework}_{g} - 2) + \gamma_{02}(\text{MixedGrade}_{g}) + \gamma_{03}(\text{Denom1v2}_{g}) + \gamma_{04}(\text{Denom1v3}_{g})$$

$$+ \gamma_{05}(\overline{\text{IQverb}}_{g}) + \gamma_{06}(\overline{\text{IQperf}}_{g}) + U_{0g}$$

$$\beta_{1g} = \gamma_{10}$$

$$\beta_{2g} = \gamma_{20}$$

```
TITLE "SAS Model 3b: Add L2 and L1 Student Performance IQ";
PROC MIXED DATA=work.StudentsSchools NOCLPRINT COVTEST NAMELEN=100 IC METHOD=REML;
     CLASS schoolID studentID;
     MODEL langpost = hw2 mixedgra den1v2 den1v3 SMIQverbz0 WSIQverbz SMIQperfz0 WSIQperfz
                      / SOLUTION DDFM=Satterthwaite OUTPM=PredIQperf;
     RANDOM INTERCEPT / TYPE=UN SUBJECT=schoolID;
     CONTRAST "Omnibus Test of Verbal IQ (Change in R2)"      SMIQverbz0 1, WSIQverbz 1;
     CONTRAST "Omnibus Test of Performance IQ (Change in R2)" SMIQperfz0 1, WSIQperfz 1;
     ESTIMATE "Denomination 2 vs 3" den1v2 -1 den1v3 1;
     ESTIMATE "L2 Contextual Effect of Verbal IQ" SMIQverbz0 1 WSIQverbz -1;
     ESTIMATE "L2 Contextual Effect of Performance IQ" SMIQperfz0  1 WSIQperfz -1;
     ESTIMATE "L2 Verbal vs Performance IQ effects"  SMIQverbz0 -1 SMIQperfz0 1;
     ESTIMATE "L1 Verbal vs Performance IQ effects"  WSIQverbz  -1 WSIQperfz  1;
     ODS OUTPUT CovParms=CovIQperf; RUN; TITLE;
```

```
display as result "STATA Model 3b: Add L2 and L1 Student Performance IQ"
mixed langpost c.hw2 c.mixedgra c.den1v2 c.den1v3 c.SMIQverbz0 ///
         c.WSIQverbz c.SMIQperfz0 c.WSIQperfz, ///
         || schoolID: , variance reml covariance(unstructured) dfmethod(satterthwaite),
       test (c.SMIQverbz0=0) (c.WSIQverbz=0), small   // Test Omnibus Verbal IQ
       test (c.SMIQperfz0=0) (c.WSIQperfz=0), small   // Test Omnibus Performance IQ
lincom c.den1v2*-1 + c.den1v3*1, small         // Denomination 2 vs 3
lincom c.SMIQverbz0*1  + c.WSIQverbz*-1, small // L1 Contextual Effect of Verbal IQ
lincom c.SMIQperfz0*1  + c.WSIQperfz*-1, small // L1 Contextual Effect of Performance IQ
lincom c.SMIQverbz0*-1 + c.SMIQperfz0*1, small // L2 Verbal vs Performance IQ Effects
lincom c.WSIQverbz*-1  + c.WSIQperf*1,  small  // L1 Verbal vs Performance IQ Effects
       predict PredIQperf, xb
       corr langpost PredIQperf
       display as result r(rho)^2 // total R2
```

### SAS Results:

Covariance Parameter Estimates

| Cov Parm | Subject | Estimate | Standard Error | Z Value | Pr > Z |
|---|---|---|---|---|---|
| UN(1,1) | schoolID | 7.5625 | 1.0831 | 6.98 | <.0001 |
| Residual | | 38.5224 | 0.9748 | 39.52 | <.0001 |

Information Criteria

| Neg2LogLike | Parms | AIC | AICC | HQIC | BIC | CAIC |
|---|---|---|---|---|---|---|
| 21703.0 | 2 | 21707.0 | 21707.0 | 21709.6 | 21713.3 | 21715.3 |

Solution for Fixed Effects

| Effect | Estimate | Standard Error | DF | t Value | Pr > \|t\| |
|---|---|---|---|---|---|
| Intercept | 40.9319 | 0.5173 | 163 | 79.13 | <.0001 |
| hw2 | 0.1075 | 0.4558 | 162 | 0.24 | 0.8138 |
| mixedgra | -1.0148 | 0.5265 | 182 | -1.93 | 0.0555 |
| den1v2 | 1.4167 | 0.5863 | 166 | 2.42 | 0.0168 |
| den1v3 | -0.1733 | 0.6146 | 169 | -0.28 | 0.7783 |
| SMIQverbz0 | 5.9262 | 0.7072 | 183 | 8.38 | <.0001 |
| WSIQverbz | 4.2895 | 0.1292 | 3123 | 33.19 | <.0001 |
| SMIQperfz0 | 2.8834 | 0.9191 | 179 | 3.14 | 0.0020 |
| WSIQperfz | 1.5946 | 0.1251 | 3123 | 12.75 | <.0001 |

## Interpret each fixed effect:

SMIQperfz0 $\gamma_{06}$ =

WSIQperfz $\gamma_{20}$ =

Estimates

| Label | Estimate | Standard Error | DF | t Value | Pr > \|t\| |
|---|---|---|---|---|---|
| Denomination 2 vs 3 | -1.5900 | 0.5986 | 167 | -2.66 | 0.0087 |
| L2 Contextual Effect of Verbal IQ | 1.6367 | 0.7189 | 196 | 2.28 | 0.0239 |
| L2 Contextual Effect of Performance IQ | 1.2888 | 0.9275 | 185 | 1.39 | 0.1664 |
| L2 Verbal vs Performance IQ effects | -3.0428 | 1.4660 | 178 | -2.08 | 0.0394 |
| L1 Verbal vs Performance IQ effects | -2.6949 | 0.2139 | 3123 | -12.60 | <.0001 |

Contrasts

| Label | Num DF | Den DF | F Value | Pr > F |
|---|---|---|---|---|
| Omnibus Test of Verbal IQ (Change in R2) | 2 | 345 | 585.84 | <.0001 |
| Omnibus Test of Performance IQ (Change in R2) | 2 | 336 | 86.17 | <.0001 |

```
* Calculate PseudoR2 relative to verbal IQ model 3a;
  %PseudoR2(NCov=2, CovFewer=CovIQverb, CovMore=CovIQperf);
```

PseudoR2 (% Reduction) for CovIQverb vs. CovIQperf

| Name | CovParm | Subject | Estimate | StdErr | ZValue | ProbZ | PseudoR2 |
|---|---|---|---|---|---|---|---|
| CovIQverb | UN(1,1) | schoolID | 7.9235 | 1.1341 | 6.99 | <.0001 | . |
| CovIQverb | Residual | | 40.5184 | 1.0252 | 39.52 | <.0001 | . |
| CovIQperf | UN(1,1) | schoolID | 7.5625 | 1.0831 | 6.98 | <.0001 | 0.045564 |
| CovIQperf | Residual | | 38.5224 | 0.9748 | 39.52 | <.0001 | 0.049261 |

```
* Calculate TotalR2 relative to verbal IQ model 3a;
  %TotalR2(DV=langpost, PredFewer=PredIQverb, PredMore=PredIQperf);
```

Total R2 (% Reduction) for PredIQverb vs. PredIQperf

| Name | Pred Corr | TotalR2 | Total R2Diff |
|---|---|---|---|
| PredIQverb | 0.62673 | 0.39279 | . |
| PredIQperf | 0.65037 | 0.42298 | 0.030191 |

## Interpret the results for pseudo-$R^2$ and total-$R^2$:

## Model 3c: Add L2 Group Mean and L1 Group-MC IQ Interactions

$$\text{Level-1: Language}_{pg} = \beta_{0g} + \beta_{1g}(\text{IQverb}_{pg} - \overline{\text{IQverb}}_{g}) + \beta_{2g}(\text{IQperf}_{pg} - \overline{\text{IQperf}}_{g}) + \beta_{3g}(\text{IQverb}_{pg} - \overline{\text{IQverb}}_{g})(\text{IQperf}_{pg} - \overline{\text{IQperf}}_{g}) + e_{pg}$$

$$\text{Level-2: } \beta_{0g} = \gamma_{00} + \gamma_{01}(\text{Homework}_{g} - 2) + \gamma_{02}(\text{MixedGrade}_{g}) + \gamma_{03}(\text{Denom1v2}_{g}) + \gamma_{04}(\text{Denom1v3}_{g}) + \gamma_{05}(\overline{\text{IQverb}}_{g}) + \gamma_{06}(\overline{\text{IQperf}}_{g}) + \gamma_{07}(\overline{\text{IQverb}}_{g})(\overline{\text{IQperf}}_{g}) + U_{0g}$$

$$\beta_{1g} = \gamma_{10}$$

$$\beta_{2g} = \gamma_{20}$$

$$\beta_{3g} = \gamma_{30}$$

```
TITLE "SAS Model 3c: Add L2 and L1 IQ Interactions";
PROC MIXED DATA=work.StudentsSchools NOCLPRINT COVTEST NAMELEN=100 IC METHOD=REML;
     CLASS schoolID studentID;
     MODEL langpost = hw2 mixedgra den1v2 den1v3 SMIQverbz0 WSIQverbz SMIQperfz0 WSIQperfz
                      SMIQverbz0*SMIQperfz0 WSIQverbz*WSIQperfz
                      / SOLUTION DDFM=Satterthwaite OUTPM=PredIQxIQ COVB;
     RANDOM INTERCEPT / TYPE=UN SUBJECT=schoolID; * COVB for regions of significance macro;
     CONTRAST "Omnibus Test of IQ Interactions (Change in R2)"
               SMIQverbz0*SMIQperfz0 1, WSIQverbz*WSIQperfz 1;
     ESTIMATE "Denomination 2 vs 3" den1v2 -1 den1v3 1;
     ESTIMATE "L2 Simple Contextual Effect of Verbal IQ"      SMIQverbz0 1 WSIQverbz -1;
     ESTIMATE "L2 Simple Contextual Effect of Performance IQ" SMIQperfz0 1 WSIQperfz -1;
     * Get simple slopes of L2 IQ perf at representative values of L2 IQ verb;
     ESTIMATE "L2 IQ perf effect at L2 IQ verbz= -1" SMIQperfz0 1 SMIQverbz0*SMIQperfz0 -1;
     ESTIMATE "L2 IQ perf effect at L2 IQ verbz=  0" SMIQperfz0 1 SMIQverbz0*SMIQperfz0  0;
     ESTIMATE "L2 IQ perf effect at L2 IQ verbz=  1" SMIQperfz0 1 SMIQverbz0*SMIQperfz0  1;
     * Get simple slopes of L1 IQ perf at representative values of L1 IQ verb;
     ESTIMATE "L1 IQ perf effect at L1 IQ verbz = -1" WSIQperfz 1 WSIQverbz*WSIQperfz -1;
     ESTIMATE "L1 IQ perf effect at L1 IQ verbz =  0" WSIQperfz 1 WSIQverbz*WSIQperfz  0;
     ESTIMATE "L1 IQ perf effect at L1 IQ verbz =  1" WSIQperfz 1 WSIQverbz*WSIQperfz  1;
     * Save info for R2 and regions of significance macro;
     ODS OUTPUT CovParms=CovIQxIQ SolutionF=FixIQxIQ CovB=CovBIQxIQ; RUN; TITLE;
```

```
display as result "STATA Model 3c: Add L2 and L1 IQ Interactions"
mixed langpost c.hw2 c.mixedgra c.den1v2 c.den1v3 c.SMIQverbz0 c.WSIQverbz c.SMIQperfz0 c.WSIQperfz ///
                c.SMIQverbz0#c.SMIQperfz0 c.WSIQverbz#c.WSIQperfz, ///
          || schoolID: , variance reml covariance(unstructured) dfmethod(satterthwaite),
       test (c.SMIQverbz0#c.SMIQperfz0) (c.WSIQverbz#c.WSIQperfz), small   // Test Omnibus Interactions
       lincom c.den1v2*-1 + c.den1v3*1, small  // Denomination 2 vs 3
// Get simple slopes of L2 IQ perf at representative values of L2 IQ verb
lincom c.SMIQperfz0*1 + c.SMIQperfz0#c.SMIQverbz0*-1, small // L2 IQ perf effect at L2 IQ verbz = -1
lincom c.SMIQperfz0*1 + c.SMIQperfz0#c.SMIQverbz0*0,  small // L2 IQ perf effect at L2 IQ verbz =  0
lincom c.SMIQperfz0*1 + c.SMIQperfz0#c.SMIQverbz0*1,  small // L2 IQ perf effect at L2 IQ verbz =  1
// Get simple slopes of L1 IQ perf at representative values of L1 IQ verb
lincom c.WSIQperfz*1 + c.WSIQperfz#c.WSIQverbz*-1, small // L1 IQ perf effect at L1 IQ verbz = -1
lincom c.WSIQperfz*1 + c.WSIQperfz#c.WSIQverbz*0,  small // L1 IQ perf effect at L1 IQ verbz =  0
lincom c.WSIQperfz*1 + c.WSIQperfz#c.WSIQverbz*1,  small // L1 IQ perf effect at L1 IQ verbz =  1
       estat vce  // Aymptotic covariance matrix of fixed effects for regions
       predict PredIQxIQ, xb
       corr langpost PredIQxIQ
       display as result r(rho)^2 // total R2
```

### SAS Results (presented in a more convenient order for interpretation):

Information Criteria

| Neg2LogLike | Parms | AIC | AICC | HQIC | BIC | CAIC |
|---|---|---|---|---|---|---|
| 21692.3 | 2 | 21696.3 | 21696.3 | 21698.9 | 21702.7 | 21704.7 |

```
                              Contrasts
                                              Num     Den
Label                                          DF      DF    F Value    Pr > F
Omnibus Test of IQ Interactions (Change in R2)  2     380       5.73    0.0035


                Covariance Parameter Estimates
                                      Standard        Z
Cov Parm      Subject      Estimate      Error    Value
UN(1,1)       schoolID       7.2581     1.0518     6.90
Residual                    38.4731     0.9737    39.51
```

Intraclass Correlation as provided by VCORR:

$$ICC = \frac{7.2581}{7.2581 + 38.4731} = .1587$$

**Design effect** using mean #students per school: = 1 + ((*n* – 1) * ICC) → 1 + [(18−1)*.1587] = **3.698**

**Effective sample size:** $N_{effective}$ = (#Total Obs) / Design Effect → 3,305 / 3.698 = **893**

**Our conditional model has reduced the design effect because a greater proportion of level-2 random intercept variance was explained (60.80%) relative to level-1 residual variance (38.56%). This means that our power to detect level-1 effects has improved—it is currently approximately that of an independent sample of 893 students (versus 677 from the empty means model).**

```
* Calculate PseudoR2 relative to empty means model 1b;
  %PseudoR2(NCov=2, CovFewer=CovEmpty, CovMore=CovIQxIQ);
```

```
PseudoR2 (% Reduction) for CovEmpty vs. CovIQxIQ
  Name      CovParm     Subject     Estimate      StdErr     ZValue     ProbZ     PseudoR2
CovEmpty    UN(1,1)     schoolID     18.5134      2.4694       7.50     <.0001      .
CovEmpty    Residual                 62.6200      1.5866      39.47     <.0001      .
CovIQxIQ    UN(1,1)     schoolID      7.2581      1.0518       6.90     <.0001     0.60795
CovIQxIQ    Residual                 38.4731      0.9737      39.51     <.0001     0.38561
```

```
* Calculate PseudoR2 relative to main effects IQ model 3b;
  %PseudoR2(NCov=2, CovFewer=CovIQperf, CovMore=CovIQxIQ);
```

```
PseudoR2 (% Reduction) for CovIQperf vs. CovIQxIQ
  Name       CovParm     Subject     Estimate      StdErr     ZValue     ProbZ     PseudoR2
CovIQperf    UN(1,1)     schoolID      7.5625      1.0831       6.98     <.0001      .
CovIQperf    Residual                 38.5224      0.9748      39.52     <.0001      .
CovIQxIQ     UN(1,1)     schoolID      7.2581      1.0518       6.90     <.0001     0.040248
CovIQxIQ     Residual                 38.4731      0.9737      39.51     <.0001     0.001282
```

```
* Calculate TotalR2 relative to main effects IQ model 3b;
  %TotalR2(DV=langpost, PredFewer=PredIQperf, PredMore=PredIQxIQ);
```

```
Total R2 (% Reduction) for PredIQperf vs. PredIQxIQ
                Pred                   Total
   Name         Corr      TotalR2      R2Diff
PredIQperf    0.65037     0.42298     .
PredIQxIQ     0.65393     0.42762     .004646174
```

**Interpret the results for pseudo-$R^2$ and total-$R^2$:**

Solution for Fixed Effects

| Effect | Estimate | Standard Error | DF | t Value | Pr > \|t\| |
|---|---|---|---|---|---|
| Intercept | 41.0988 | 0.5120 | 166 | 80.27 | <.0001 |
| hw2 | 0.1549 | 0.4491 | 161 | 0.34 | 0.7306 |
| mixedgra | -0.8030 | 0.5265 | 183 | -1.53 | 0.1289 |
| den1v2 | 1.4829 | 0.5784 | 165 | 2.56 | 0.0112 |
| den1v3 | -0.09376 | 0.6059 | 168 | -0.15 | 0.8772 |
| SMIQverbz0 | 5.3809 | 0.7269 | 177 | 7.40 | <.0001 |
| WSIQverbz | 4.2784 | 0.1293 | 3123 | 33.10 | <.0001 |
| SMIQperfz0 | 2.8136 | 0.9058 | 177 | 3.11 | 0.0022 |
| WSIQperfz | 1.6068 | 0.1251 | 3123 | 12.84 | <.0001 |
| SMIQverbz0*SMIQperfz0 | -2.2914 | 0.9280 | 203 | -2.47 | 0.0144 |
| WSIQverbz*WSIQperfz | -0.2806 | 0.1197 | 3205 | -2.34 | 0.0192 |

## Interpret each fixed effect:

SMIQverbz0 $\gamma_{05}$ =

SMIQperfz0 $\gamma_{06}$ =

SMIQverbz0 *SMIQperfz0 $\gamma_{07}$ =

WSIQverbz $\gamma_{10}$ =

WSIQperfz $\gamma_{20}$ =

WSIQverbz*WSIQperfz $\gamma_{30}$ =

Estimates

| Label | Estimate | Standard Error | DF | t Value | Pr > \|t\| |
|---|---|---|---|---|---|
| Denomination 2 vs 3 | -1.5766 | 0.5894 | 166 | -2.68 | 0.0082 |
| L2 Simple Contextual Effect of Verbal IQ | 1.1026 | 0.7382 | 188 | 1.49 | 0.1370 |
| L2 Simple Contextual Effect of Performance IQ | 1.2068 | 0.9144 | 184 | 1.32 | 0.1885 |
| L2 IQ perf effect at L2 IQ verb z = -1 | 5.1051 | 1.2713 | 198 | 4.02 | <.0001 |
| L2 IQ perf effect at L2 IQ verb z = 0 | 2.8136 | 0.9058 | 177 | 3.11 | 0.0022 |
| L2 IQ perf effect at L2 IQ verb z = 1 | 0.5222 | 1.3219 | 183 | 0.40 | 0.6933 |
| L1 IQ perf effect at L1 IQ verb z = -1 | 1.8874 | 0.1767 | 3167 | 10.68 | <.0001 |
| L1 IQ perf effect at L1 IQ verb z = 0 | 1.6068 | 0.1251 | 3123 | 12.84 | <.0001 |
| L1 IQ perf effect at L1 IQ verb z = 1 | 1.3262 | 0.1696 | 3163 | 7.82 | <.0001 |

Covariance Matrix for Fixed Effects needed for Regions goes here, omitted to save space

```
* Calculate regions of significance for effect of L2 IQperf;
  %Regions(FixData=FixIQxIQ,CovBData=CovBIQxIQ,Pred=SMIQperfz0,
           Mod=SMIQverbz0,ModCenter=0,Interact=SMIQverbz0*SMIQperfz0,Order=9);
```

**Regions of significance for SMIQverbz0*SMIQperfz0 interaction:**
The effect of SMIQperfz0 will be significant at centered values of SMIQverbz0 BELOW the lower bound and ABOVE the upper bound, which translate to these uncentered lower and upper bounds.

| Centered Lower | Centered Upper | Uncentered Lower | Uncentered Upper |
|---|---|---|---|
| 0.38421 | 6.38428 | 0.38421 | 6.38428 |

**Level-2 Regions of Significance Result:** This means that the school-level effect of performance IQ will be significantly positive for school mean verbal IQ z ≤ 0.38, nonsignificant for schools between z = 0.38 and z= 6.38, and significantly negative for school mean verbal IQ z ≥ 6.38. Thus, for a little over half of the schools (school mean verbal IQ z ≤ 0.38), having greater mean performance IQ than other schools is predicted to help their mean language. For the rest of the schools (school mean verbal IQ z > 0.38), greater school mean performance IQ than other schools is not predicted to help their mean language.

```
* Calculate regions of significance for effect of L1 IQperf;
  %Regions(FixData=FixIQxIQ,CovBData=CovBIQxIQ,Pred=WSIQperfz,
           Mod=WSIQverbz,ModCenter=0,Interact= WSIQverbz*WSIQperfz,Order=10);
```

```
Regions of significance for WSIQverbz*WSIQperfz interaction:

The effect of WSIQperfz will be significant at centered values of WSIQverbz BELOW the lower bound
and ABOVE the upper bound, which translate to these uncentered lower and upper bounds.

Centered    Centered    Uncentered    Uncentered
  Lower       Upper        Lower         Upper
 3.05751     34.8995     3.05751       34.8995
```

**Level-1 Regions of Significance Result:** This means that the student-level effect of performance IQ will be significantly positive for within-school verbal IQ z ≤ 3.06, nonsignificant for students between z = 3.06 and z = 34.90, and significantly negative for within-school verbal IQ z ≥ 34.90. Thus, for the vast majority of students (within-school verbal IQ z ≤ 3.06), having greater performance IQ than other students in their school is predicted to help their language. Only for very few students (within-school verbal IQ z > 3.06) is greater performance IQ than other students in their school not predicted to help their language.

---

## Sample Results Section [indicates notes about what to customize or also include]

The extent to which the extent to which student language at post-test could be predicted from school-level variables of homework amount, denomination (three types), and use of mixed-grade classes, as well as student-level predictors of verbal IQ and performance IQ, was examined in a series of multilevel models in which the 3,305 students were modeled as nested within their 178 schools. Restricted Maximum likelihood (REML) within SAS [or STATA] MIXED was used in estimating and reporting all model parameters. The significance of fixed effects was evaluated with univariate and multivariate Wald using Satterthwaite denominator degrees of freedom, whereas random effects were evaluated via likelihood ratio tests (i.e., −2ΔLL with degrees of freedom equal to the number of new random effects variances and covariances). Alpha was chosen as .05. Model-implied fixed effects were requested via ESTIMATE [or LINCOM] statements. Effect size for the fixed effects was evaluated via pseduo-$R^2$ values for the proportion reduction in each variance component, as well as with total-$R^2$, the squared correlation between actual language and language as predicted by the model fixed effects.

As derived from an empty means, random intercept model, student language had an intraclass correlation of ICC = .228, indicating that 22.8% of the variance in student language was between schools, a significant amount, −2ΔLL(1) = 436.79, $p$ < .001. We first added the school-level predictors of homework amount (centered at 2), denomination (three types; Type 1 was the reference), and use of mixed-grade classes (none as the reference) to the model predicting the level-2 random intercept. These four fixed effects accounted for significant variance overall, $F$(4, 162) = 6.68, $p$ < .001, including 13.8% of the level-2 random intercept variance and 3.21% of the total variance. The variance accounted for remained significant after controlling for the effect of homework amount (which was not significant), $F$(3, 164) = 8.28, $p$ < .001. Schools that used mixed-grade classes had significantly lower average language (Est = −2.01, SE = 0.73). The effect of denomination was significant as well, $F$(2, 159) = 6.61, $p$ < .002. Type 2 schools had significantly higher language than both Type 1 schools (Est = 2.74, SE = 0.81) and Type 3 schools (Est = 2.27, SE = 0.83), which did not differ. We then examined an interaction between the use of mixed-grade classes and denomination, but given that none of the interaction contrasts were significant, $F$(2, 172) = 1.26, $p$ = .287, change in pseudo-$R^2$ < .004, this interaction was removed.

We then examined the sources of variance in the student-level IQ predictors (which were each z-scored to create comparable scales). An empty means, random intercept model indicated ICC = .120 for verbal IQ and ICC = .048 for performance IQ; both ICCs were significant (−2ΔLL $p$'s < .001). Thus, we included IQ predictor effects at both levels—the school mean of IQ captured the between-school effect at level 2, and student IQ centered at the school mean captured the within-school effect at level 1. We first added effects at both levels for verbal IQ, which contributed significantly to the model overall, $F(2, 361) = 931.48$, $p < .001$, change in total-$R^2 = .361$; each effect was significant. At level 2, schools with higher mean student verbal IQ than other schools had higher language on average (Est = 7.30, SE = 0.57), which accounted for an additional 50.3% of the remaining level-2 random intercept variance. At level 1, students with higher verbal IQ than other students in their school also had higher language than other students in their school (Est = 4.97, SE = 0.12), which accounted for 35.3% of the remaining level-1 residual variance. The contextual level-2 effect (Est = 2.33, SE = 0.58) indicated that the level-2 school verbal IQ effect was significantly greater than the level-1 student verbal IQ effect (i.e., a significantly positive incremental effect of school mean verbal IQ after controlling for student verbal IQ). With respect to the previous predictors, the effect of homework amount remained nonsignificant, the advantage for Type 2 schools remained significant, and the effect of using mixed-grade classes became nonsignificant.

We then added effects at both levels for performance IQ, which contributed significantly to the model overall, $F(2, 336) = 86.17$, $p < .001$, change in total-$R^2 = .031$; each effect was significant. At level 2, schools with higher mean student performance IQ than other schools had higher language on average (Est = 2.88, SE = 0.92), which accounted for an additional 4.56% of the remaining level-2 random intercept variance. At level 1, students with higher performance IQ than other students in their school also had higher language than other students in their school (Est = 1.59, SE = 0.13), which accounted for 4.93% of the remaining level-1 residual variance. The contextual level-2 effect (Est = 1.29, SE = 0.93) was not significant, indicating equivalent effects of level-2 and level-1 performance IQ (i.e., no significant incremental effect of school mean performance IQ after controlling for student performance IQ). The unique effects of verbal IQ remained significant (level-2: Est = 5.93, SE = 0.71; level-1: Est = 4.29, SE = 0.13) and were also significantly larger than the unique effects of performance IQ (level-2 difference: Est = 3.04, SE = 1.47; level-1 difference: Est = 2.69, SE = 0.21). With respect to the previous predictors, the effect of homework amount remained nonsignificant, the advantage for Type 2 schools remained significant, and the effect of using mixed-grade classes remained nonsignificant.

Finally, we considered the interaction between verbal IQ and performance IQ at each level. This final model is given in [provide Equation for Model 3c] and results are shown in Table X [table should provide estimates, SEs, and *p*-values (or stars or bold font) for all model parameters, including variance parameters]. The final model accounted for 60.8% of the level-2 random intercept variance, 38.6% of the level-1 residual variance, and 42.8% of the overall language variance. With respect to the previous predictors, the effect of homework amount remained nonsignificant, the advantage for Type 2 schools remained significant, and the effect of using mixed-grade classes remained nonsignificant. The two new IQ interactions which contributed significantly to the model overall, $F(2, 380) = 5.73$, $p < .004$, change in total-$R^2 = .005$; each interaction was significant. Regions of significance were then calculated to describe the range of simple effects of performance IQ implied by these interactions. First, at level 2, the effect of school mean performance IQ (of Est = 2.81, SE = 0.91 for school mean verbal IQ = 0) became significantly weaker (less positive) by 2.29 (SE = 0.93) for every unit higher school mean verbal IQ, which accounted for 4.02% of the remaining level-2 random intercept variance. Regions of significance for the level-2 interaction indicated that the school-level effect of performance IQ will be significantly positive for schools with mean verbal IQ $z \leq 0.38$, nonsignificant for schools between $z = 0.38$ and $z = 6.38$, and significantly negative for schools with mean verbal IQ $z \geq 6.38$. Thus, for a little over half of the schools (school mean verbal IQ $z \leq 0.38$), having greater mean performance IQ than other schools is predicted to help their mean language. For the rest of the schools (school mean verbal IQ $z > 0.38$), greater school mean performance IQ than other schools is not predicted to help their mean language. Next, at level 1, the effect of student performance IQ (of Est = 1.61, SE = 0.13 at within-school student verbal IQ = 0) also became significantly weaker (less positive) by 0.29 (SE = 0.12) for every unit higher within-school student verbal IQ, which accounted for .001% of the remaining level-1 residual variance. Regions of significance for the level-1 interaction indicated that student-level effect of performance IQ will be significantly positive for within-school verbal IQ $z \leq 3.06$, nonsignificant for students between $z = 3.06$ and $z = 34.90$, and significantly negative for within-school verbal IQ $z \geq 34.90$. Thus, for the vast majority of students (within-school verbal IQ $z \leq 3.06$), having greater performance IQ than other students in their school is predicted to help their language. Only for very few students (within-school verbal IQ $z > 3.06$) is greater performance IQ than other students in their school not predicted to help their language.